
Введение
Интеллектуальный анализ данных как одно из новых междисциплинарных направлений включает в себя различные способы работы сданными и в зависимости от предметной области имеет некоторую специфику применения этих способов.
Ставшее уже классическим data mining применимо не ко всем задачам технологии транспортных процессов. В частности, исследование транспортных потоков требует специфических подходов к очистке, агрегированию, преобразованию и последующему анализу данных о потоке.
Знания, полученные при изучении «Математических методов в транспортных исследованиях», в полной мере могут быть применены при интеллектуальном анализе данных о транспортных потоках.
[bookmark: _GoBack] Последовательное выполнение практических заданий по дисциплине «Интеллектуальный анализ данных на транспорте» позволит получить необходимые магистранту навыки научных исследований и компетенции.
Для выполнения этого задания следует получить первичные данные у преподавателя или на сайте do.skif.donstu.ru  в соответствии со своим вариантом. Номер варианта задания равен номеру студента в списке группы.

Задание №1
Согласно данным двадцати (п = 20) детекторов, требуется построить линейную регрессионную модель скорости на основе следующих показателей: у — скорость; xi — число транспортных средств i-го типа. Исходные данные для анализа приведены на Листе 1 файла с исходными данными.
Алгоритм выполнения задания приведен на рисунке 1. Для проверки значимости коэффициентов регрессии использовать критерий Стьюдента с вероятностью 0,1 и числом степеней свободы n-k, где k- количество коэффициентов в уравнении регрессии. Для проверки значимости уравнения регрессии использовать критерий Фишера с вероятностью 0,1 и числом степеней свободы k и n-k. 


Ввод исходных данных
Получить уравнение регрессии
Есть не значимые коэффициенты?
Проверить значимость коэффициентов
Получить уравнение регрессии для значимых и не коррелированных факторов
Проверить значимость уравнения
Вывод 
Получить матрицу парных коэффициентов корреляции
нет
да
























Рисунок 1. Алгоритм выполнения задания 1.



Задание №2
Применить алгоритм кластеризации K-means.
Выбрать количество центроидов.
1. Выбрать центроиды
2. Определить расстояние между точками и центроидами.
=КОРЕНЬ(($B3-H$3)^2+($C3-H$4)^2+($D3-H$5)^2+($E3-H$6)^2)
3. Определить , к какому кластеру относится каждая категория
=ЕСЛИ(МИН(M3:P3)=M3;"C1";ЕСЛИ(МИН(M3:P3)=N3;"C2";ЕСЛИ(МИН(M3:P3)=O3;"C3";"C4")))
4. Определить новые центроиды.
5. Проверить, сместились ли центроиды? Если не сместились, то последнее распределение по кластерам – это результат . Если сместились, то  идем в п.2.
Исходные данные для анализа приведены на Листе 2 файла с исходными данными.

Задание №3
Применить алгоритм кластеризации kNN.

1. Рассчитать расстояние distance до соседей, используя Евклидову метрику.
=((B2-$G$2)^2+(C2-$H$2)^2)^0,5
2. Определить параметр small для k (k=1,…,11) ближайших соседей.
=НАИМЕНЬШИЙ(E$2:E$201;F4)
3.  Определить, сколько из k ближайших соседей имеют Response=1, а сколько Response=0.
=СЧЁТЕСЛИМН($D$2:$D$201;J$3;$E$2:$E$201;"<="&$I4)
=СЧЁТЕСЛИМН($D$2:$D$201;K$3;$E$2:$E$201;"<="&$I4)
4. Определить, параметр «Daisy's likely response», сравнивая результаты выбора ближайших соседей.
=ЕСЛИ(J4>K4;$J$3;$K$3)
Исходные данные для анализа приведены на Листе 3 файла с исходными данными.
Задание №4
Построить дерево решений.
Исходные данные о погодных условиях и проведении игры приведены на Листе 4 файла с исходными данными.  Построить дерево решений о том, будет ли назначена игра при том или ином сочетании погодных условий?

Введите “Sample-size” в ячейку A16 и 14 в ячейку B16. 
14 - это объем данных.
Введите “p∗log(p)” в ячейку E17.  p∗log(p) представляет PKlog2(Pk) . Введите “Entropy” в ячейку F17. Объедините ячейки B18 и B19 вместе и введите “Play” внутри объединенной ячейки. Введите “yes” в ячейку C18 и “no” в ячейку C19.
Для вычисления энтропии для целевой переменной Play:
В ячейке D18 введите формулу, которая  подсчитывает, сколько "yes" в столбце Play:
=COUNTIFS($E$2:$E$15;C18). Или =СЧЁТЕСЛИМН(E3:E16;C19)
Автозаполнение из ячейки D18 в ячейку D19. D19 подсчитывает, сколько “no” в столбце Play. В ячейке E18 введите формулу =D18/$B$16*LOG(D18/$B$16;2).  Эта формула вычисляет Pyes = Log2(Pyes). Автоматическое заполнение с E18 на E19.  E19 вычисляет Pno = Log2(Pno). В F18 ведите формулу =−SUM(E18:E19). Объедините ячейки F18 и F19.
Результатом является энтропия для целевой переменной Play (рисунок 1).

Рисунок 1. Строки с 16 по 19[image: ]

Дальше нужно будет использовать функцию ИНДЕКС (или INDEX). 
ИНДЕКС ($A$2:$E$15,0,1) извлекает столбец температуры, ИНДЕКС($A$2:$E$15,0,4) извлекает столбец Outlook.
Введите вспомогательные числа 1, 1, 1, 2, 2, 3, 3, 4, 4, и 4 в ячейки A22:A31 соответственно.
Введите “yes”, “no”, “p*log(p)-yes”, “p*log(p)-no”, “weighted”, “entropy” и “info gain” в ячейки D21:J21 соответственно. 
Столбец “weighted” хранит взвешенную энтропию для каждого значения атрибута.
Заполните ячейки A22:C31 как показано на рисунке 2.
[image: ]
Рисунок 2. Образец заполнения ячеек A22:C31

Для вычисления энтропии и прироста информации введите следующую формулу в ячейку D22:
=COUNTIFS(INDEX($A$2:$E$15;0;$A22);$C22;$E$2:$E$15;D$21)
или
=СЧЁТЕСЛИМН(ИНДЕКС($A$2:$E$15;0;$A22);$C22;$E$2:$E$15;D$21)
Ячейка A22 ссылается на 1; следовательно, ИНДЕКС($A$2:$E$15,0,$A22) извлекает столбец температуры. Эта формула подсчитывает, сколько существует точек данных, температура которых является “hot”, а Play= “yes” одновременно.
Автоматическое заполнение ячеек D22-E22, а затем автоматическое совместное заполнение ячеек D31:E31. 
Введите следующую формулу в ячейку F22:
=IF(ISERROR(D22/SUM($D22:$E22) * LOG(D22/SUM($D22:$E22),2)),0,D22/SUM($D22:$E22)*LOG(D22/SUM($D22:$E22),2))
или
=ЕСЛИ(ЕОШИБКА(D22/СУММ($D22:$E22)*LOG(D22/СУММ($D22:$E22);2));0;D22/СУММ($D22:$E22)*LOG(D22/СУММ($D22:$E22);2))
Эта формула вычисляет Pyes × Log2(Pyes) для значения атрибута “hot”.
Если  SUM($D 22:$E 22) возвращает 0, то в выражении будет ошибка деления на ноль. Кроме того, функция log не может принимать 0 в качестве входных данных. Таким образом, функция ISERROR ( или ЕОШИБКА) используется здесь для перехвата таких ошибок. Если возникает ошибка, возвращается 0. Использование функции IF вместе с ISERROR здесь имеет решающее значение.
Автоматическое заполнение из ячейки F22 в ячейку G22, а затем автоматическое совместное заполнение в ячейки F31:G31. Ячейка G22 вычисляет Pno × Log2(Pno) для значения атрибута ”hot".
В ячейке H22 введите формулу 
 =−SUM($D22:$E22)/$B$16*(F22+G22) 
или
=-СУММ($D22:$E22)/$B$16*(F22+G22)
Вычисленное значение представляет собой взвешенную субэнтропию для значения “hot” атрибута Temperature.  Автоматическое заполнение из ячейки H22 в ячейку H31.
В ячейке I22 введите формулу  =SUMIFS(H$22:H$31,$A$22: A$31,A 22)  или
=СУММЕСЛИМН(H$22:H$31;$A$22:A$31;A23) и выполните автозаполнение из ячейки I22 в I31.
 Формула в ячейке I22 вычисляет энтропию для атрибута Temperature.
В ячейке J22 введите формулу =F$18-I22, чтобы получить прирост информации для атрибута Temperature. 
Автоматическое заполнение из ячейки J22 в ячейку J31.  Объедините ячейки I22:I24, I25:I26, I27:I28, I29:I31, J22:J24, J25:J26, J27:J28 и J29:J31 соответственно.
Вычисление завершено(рисунок 3). Поскольку атрибут Outlook имеет наибольший информационный прирост, он выбран для разделения узла дерева уровня 1.

Рисунок 3. Результат вычисления энтропии и прироста информации[image: ]

Поскольку атрибут Outlook имеет наибольший информационный прирост, он выбран для разделения узла дерева уровня 1.  Поскольку взвешенная энтропия H-outlook-overcast равна нулю, дочерний узел overcast(4,0) является конечным узлом - листом.  Мы можем нарисовать простую “древовидную диаграмму”, как показано на рисунке 4. 
[image: ]

Рисунок 4. Простое изображение дерева решений

Следующая задача - разделить узлы rainy(3,2) и sunny(2,3). 
Дальше следует разделить узел rainy(3,2).
Создайте копию рабочего листа level-1 и переименуйте новый рабочий лист level-2-rainy. На рабочем листе level-2-rainy введите “rainy” в ячейку F1.
В ячейке B16 введите формулу  =COUNTIF($D$2:$D$15,$F$1)  или =СЧЁТЕСЛИ($D$2:$D$15;$F$1) 
При этом будут учитываться только те ячейки, прогнозное значение которых равно “rainy”.
Измените формулу в ячейке D18, добавив два параметра:
=СЧЁТЕСЛИМН($E$2:$E$15;C18;$D$2:$D$15;$F$1)
или
=COUNTIFS($E$2:$E$15;C18;$D$2:$D$15;$F$1). 
Автоматическое заполнение из ячейки D18 в ячейку D19.
Вставьте ";$D$2:$D$15;$F$1" в формулу ячейки D22 и убедитесь, что ее формула выглядит так:
=СЧЁТЕСЛИМН(ИНДЕКС($A$2:$E$15;0;$A22);$C22;$E$2:$E$15;D$21;$D$2:$D$15;$F$1)
или
=COUNTIFS(INDEX($A$2:$E$15;0;$A22);$C22;$E$2:$E$15;D$21;$D$2:$D$15;$F$1)
Автозаполнение из D22 в E22, а затем автозаполнение вместе в ячейки D31:E31. В диапазоне F22:J31 все значения пересчитаются автоматически.
Максимальный прирост информации у атрибута Windy. Узел rainy(3,2) должен быть разветвлен на основе Windy.Поскольку два дочерних узла Windy-t(0,2) и Windy-f(3,0) имеют нулевую энтропию, они оба являются конечными узлами - листьями.  Результат на рисунке 5.
[image: ]
Рисунок 5. Второй уровень дерева решений

Разделить узел sunny(2,3).
Сделайте копию рабочего листа level-2-rainy, переименуйте новый рабочий лист level-2-sunny.
На рабочем листе level-2-sunny измените текст внутри ячейки F1 на “sunny”. Результат виден сразу.
Из результатов, показанных на рабочем листе level-2-sunny, видно, что древовидный узел sunny(2,3) должен быть разветвлен в зависимости от влажности. Поскольку оба сгенерированных дочерних узла имеют нулевую энтропию, они оба являются конечными узлами. Таким образом, больше нет необходимости разделять дерево. 
Последние два листа дерева решений показать на диаграмме самостоятельно.
Задание №5
Применить метод ассоциативного анализа к исходным данным о наборах объектов
Формулы для выполнения задания:

	Ячейка
	Формула
	Диапазон автозаполне-ния

	D2
	="I"&C2
	D2:D3539

	F2
	=СЧЁТЕСЛИМН($A$2:$A$3539,$E2,$D$2:$D$3539,F$1)
	F2:BC1001

	BG2
	=СЧЁТЕСЛИМН(F2:F1001,1) 
	BG2:DD2

	BG3
	=СЧЁТЕСЛИМН(ИНДЕКС($F$2:$BC$1001;0;$BE3);1;F$2:F$1001;1)
	BG3:DD52

	DH3
	=ЕСЛИ($DG3=DH$1,"",ЕСЛИ(И(BG$2>0,BG3/1000>=$DF$3,BG3/BG$2>=$DF$5),ТЕКСТ(BG3/1000,"0.000")&";"&ТЕКСТ(BG3/BG$2,"0.000"),""))
	DH3:FE52


Найти такие пороговые значения Support и Confidence, при которых не найдется ни одной пары объектов. 


Задание №6
Применить метод «Наивный Байес» к нечисловым характеристикам объектов исследования
Формулы для выполнения задания:
	Ячейка
	Формула
	Диапазон автозаполне-ния

	C3
	=СЧЁТЕСЛИ($E$13:$E$22;C$2)
	C3:D3

	C4
	=C$3/$B1
	C4:D4

	C5
	=СЧЁТЕСЛИМН($E$13:$E$22;C$2;$B$13:$B$22;$B5)/C$3
	C5:D6

	C7
	=СЧЁТЕСЛИМН($E$13:$E$22;C$2;$C$13:$C$22;$B7)/C$3
	C7:D8

	C9
	=СЧЁТЕСЛИМН($E$13:$E$22;C$2;$D$13:$D$22;$B9)/C$3
	C9:D10

	F23
	=СУММЕСЛИ($B$5:$B$10;$B23;C$5:C$10)*СУММЕСЛИ($B$5:$B$10;$C23;C$5:C$10)*СУММЕСЛИ($B$5:$B$10;$D23;C$5:C$10)*C$4
	F23:G23

	H23
	=F23/(F23+G23)
	H23

	E23
	=ЕСЛИ(H23>0,5;"yes";"no")
	E23



ЛИТЕРАТУРА

1. Learn Data Mining Through Excel: A Step-by-Step Approach for Understanding Machine Learning Methods.  Hong Zhou. Режим доступа: https://doi.org/10.1007/978-1-4842-5982-5 
2. Data mining : practical machine learning tools and techniques / Ian H. Witten, Eibe  Frank. -: Morgan Kaufmann Publishers, 2020. - 525 p.

25 |high 3 4 -0.52388247 -0.46135 0.492614 0.78845  0.151836
26 \normal 6 1 -0.19062208 -0.40105 0.295836
27| TRUE 3 3 -0.5 -0.5 0.428571 0.892159 0.048127
28| FALSE 6 2 -0.31127812  -0.5 0.463587
29 overcast 4 0 0 0 0
30 |rainy 3 2 -0.44217936 -0.52877 0.346768 0.693536 0.24675
31 |sunny 2 3 -0.52877124 -0.44218 0.346768




